
pkgname=jfsutils
pkgver=1.1.15
pkgrel=5
pkgdesc="JFS filesystem utilities"
arch=('x86_64')
url="http://jfs.sourceforge.net"
license=('GPL')
groups=('base')
depends=('e2fsprogs' 'util-linux')
source=("http://jfs.sourceforge.net/project/pub/${pkgname}-${pkgver}.tar.gz"
        'inttypes.patch')
md5sums=('8809465cd48a202895bc2a12e1923b5d'
         '444078802d798606be9e32fda7cf7376')

build() {
  cd ${pkgname}-${pkgver}
  patch -p1 -i ${srcdir}/inttypes.patch
  
  ./configure --prefix=/usr --sbindir=/sbin 
  make
}

check() {
  cd ${pkgname}-${pkgver}
  
  make check
}

package() {
  cd ${pkgname}-${pkgver}
  
  make DESTDIR=${pkgdir} install
}

